

## Programming Languages and Compilers (CS 421)

Elsa L Gunter  
2112 SC, UIUC

<http://courses.engr.illinois.edu/cs421>


Based in part on slides by Mattox Beckman, as updated  
by Vikram Adve and Gul Agha

10/18/12

1

## Disjoint Union Types

- Disjoint union of types, with some possibly occurring more than once



- We can also add in some new singleton elements

10/18/12

2

## Disjoint Union Types

```
# type id = DriversLicense of int
| SocialSecurity of int | Name of string;;
type id = DriversLicense of int | SocialSecurity
of int | Name of string
# let check_id id = match id with
  DriversLicense num ->
    not (List.mem num [13570; 99999])
  | SocialSecurity num -> num < 900000000
  | Name str -> not (str = "John Doe");;
val check_id : id -> bool = <fun>
```

10/18/12

3

## Polymorphism in Variants

- The type 'a option gives us something to represent non-existence or failure

```
# type 'a option = Some of 'a | None;;
type 'a option = Some of 'a | None
```

- Used to encode partial functions
- Often can replace the raising of an exception

10/18/12

4

## Functions producing option

```
# let rec first p list =
  match list with [ ] -> None
  | (x::xs) -> if p x then Some x else first p xs;;
val first : ('a -> bool) -> 'a list -> 'a option = <fun>
# first (fun x -> x > 3) [1;3;4;2;5];;
- : int option = Some 4
# first (fun x -> x > 5) [1;3;4;2;5];;
- : int option = None
```

10/18/12

5

## Functions over option

```
# let result_ok r =
  match r with None -> false
  | Some _ -> true;;
val result_ok : 'a option -> bool = <fun>
# result_ok (first (fun x -> x > 3) [1;3;4;2;5]);;
- : bool = true
# result_ok (first (fun x -> x > 5) [1;3;4;2;5]);;
- : bool = false
```

10/18/12

6

## Folding over Variants


```
# let optionFold someFun noneVal opt =
  match opt with None -> noneVal
  | Some x -> someFun x;;
val optionFold : ('a -> 'b) -> 'b -> 'a option ->
  'b = <fun>
# let optionMap f opt =
  optionFold (fun x -> Some (f x)) None opt;;
val optionMap : ('a -> 'b) -> 'a option -> 'b
  option = <fun>
```

10/18/12

7

## Recursive Types

- The type being defined may be a component of itself



10/18/12

8

## Mapping over Variants

```
# let optionMap f opt =
  match opt with None -> None
  | Some x -> Some (f x);;
val optionMap : ('a -> 'b) -> 'a option -> 'b
  option = <fun>
# optionMap
  (fun x -> x - 2)
  (first (fun x -> x > 3) [1;3;4;2;5]);;
- : int option = Some 2
```

10/18/12

9

## Recursive Data Types

```
# type int_Bin_Tree =
  Leaf of int | Node of (int_Bin_Tree *
    int_Bin_Tree);;
```

```
type int_Bin_Tree = Leaf of int | Node of
  (int_Bin_Tree * int_Bin_Tree)
```

10/18/12

10

## Recursive Data Type Values


```
# let bin_tree =
  Node(Node(Leaf 3, Leaf 6), Leaf (-7));;
```

```
val bin_tree : int_Bin_Tree = Node (Node
  (Leaf 3, Leaf 6), Leaf (-7))
```

10/18/12

11

## Recursive Data Type Values



10/18/12

12

## Recursive Functions

```
# let rec first_leaf_value tree =  
  match tree with (Leaf n) -> n  
  | Node (left_tree, right_tree) ->  
    first_leaf_value left_tree;;  
val first_leaf_value : int_Bin_Tree -> int =  
  <fun>  
# let left = first_leaf_value bin_tree;;  
val left : int = 3
```

10/18/12

13

## Mapping over Recursive Types

```
# let rec ibtreeMap f tree =  
  match tree with (Leaf n) -> Leaf (f n)  
  | Node (left_tree, right_tree) ->  
    Node (ibtreeMap f left_tree,  
          ibtreeMap f right_tree);;  
val ibtreeMap : (int -> int) -> int_Bin_Tree ->  
  int_Bin_Tree = <fun>
```

10/18/12

14

## Mapping over Recursive Types

```
# ibtreeMap ((+) 2) bin_tree;;  
  
- : int_Bin_Tree = Node (Node (Leaf 5, Leaf  
  8), Leaf (-5))
```

10/18/12

15

## Folding over Recursive Types

```
# let rec ibtreeFoldRight leafFun nodeFun tree =  
  match tree with Leaf n -> leafFun n  
  | Node (left_tree, right_tree) ->  
    nodeFun  
      (ibtreeFoldRight leafFun nodeFun left_tree)  
      (ibtreeFoldRight leafFun nodeFun right_tree);;  
val ibtreeFoldRight : (int -> 'a) -> ('a -> 'a -> 'a) ->  
  int_Bin_Tree -> 'a = <fun>
```

10/18/12

16

## Folding over Recursive Types

```
# let tree_sum =  
  ibtreeFoldRight (fun x -> x) (+);;  
val tree_sum : int_Bin_Tree -> int = <fun>  
# tree_sum bin_tree;;  
- : int = 2
```

10/18/12

17

## Mutually Recursive Types

```
# type 'a tree = TreeLeaf of 'a  
  | TreeNode of 'a treeList  
and 'a treeList = Last of 'a tree  
  | More of ('a tree * 'a treeList);;  
type 'a tree = TreeLeaf of 'a | TreeNode of 'a  
  treeList  
and 'a treeList = Last of 'a tree | More of ('a  
  tree * 'a treeList)
```

10/18/12

18

## Mutually Recursive Types - Values

```
# let tree =
  TreeNode
    (More (TreeLeaf 5,
      (More (TreeNode
        (More (TreeLeaf 3,
          Last (TreeLeaf 2))),
        Last (TreeLeaf 7)))));;
```

10/18/12

19


## Mutually Recursive Types - Values

```
val tree : int tree =
  TreeNode
    (More
      (TreeLeaf 5,
        More
          (TreeNode (More (TreeLeaf 3, Last
            (TreeLeaf 2))), Last (TreeLeaf 7)))))
```

10/18/12

20

## Mutually Recursive Types - Values




10/18/12

21

## Mutually Recursive Types - Values

A more conventional picture



10/18/12

22

## Mutually Recursive Functions

```
# let rec fringe tree =
  match tree with (TreeLeaf x) -> [x]
  | (TreeNode list) -> list_fringe list
and list_fringe tree_list =
  match tree_list with (Last tree) -> fringe tree
  | (More (tree, list)) ->
    (fringe tree) @ (list_fringe list);;
```

```
val fringe : 'a tree -> 'a list = <fun>
val list_fringe : 'a treeList -> 'a list = <fun>
```

10/18/12

23

## Mutually Recursive Functions

```
# fringe tree;;
- : int list = [5; 3; 2; 7]
```

10/18/12

24

## Nested Recursive Types

```
# type 'a labeled_tree =
  TreeNode of ('a * 'a labeled_tree
    list);;
type 'a labeled_tree = TreeNode of ('a
  * 'a labeled_tree list)
```

10/18/12

25

## Nested Recursive Type Values

```
# let ltree =
  TreeNode(5,
    [TreeNode(3, []);
     TreeNode(2, [TreeNode(1, []);
                    TreeNode(7, [])]);
     TreeNode(5, [])]);;
```

10/18/12

26


## Nested Recursive Type Values

```
val ltree : int labeled_tree =
  TreeNode
    (5,
    [TreeNode(3, []); TreeNode(2,
    [TreeNode(1, []); TreeNode(7, [])]);
     TreeNode(5, [])])
```

10/18/12

27


## Nested Recursive Type Values



10/18/12

28

## Nested Recursive Type Values



10/18/12

29

## Mutually Recursive Functions

```
# let rec flatten_tree labtree =
  match labtree with TreeNode(x, treelist)
    -> x::flatten_tree_list treelist
  and flatten_tree_list treelist =
  match treelist with [] -> []
  | labtree::labtrees
    -> flatten_tree labtree
      @ flatten_tree_list labtrees;;
```

10/18/12

30

## Mutually Recursive Functions

```
val flatten_tree : 'a labeled_tree -> 'a list =  
  <fun>  
val flatten_tree_list : 'a labeled_tree list -> 'a  
  list = <fun>  
# flatten_tree ltree;;  
- : int list = [5; 3; 2; 1; 7; 5]
```

- Nested recursive types lead to mutually recursive functions

10/18/12

31

## Infinite Recursive Values

```
# let rec ones = 1::ones;;  
val ones : int list =  
  [1; 1; 1; 1; ...]  
# match ones with x::_ -> x;;  
Characters 0-25:  
Warning: this pattern-matching is not exhaustive.  
Here is an example of a value that is not matched:  
[]  
  match ones with x::_ -> x;;  
  ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^  
- : int = 1
```

10/18/12

32

## Infinite Recursive Values

```
# let rec lab_tree = TreeNode(2, tree_list)  
  and tree_list = [lab_tree; lab_tree];;  
val lab_tree : int labeled_tree =  
  TreeNode (2, [TreeNode(...); TreeNode(...)])  
val tree_list : int labeled_tree list =  
  [TreeNode (2, [TreeNode(...);  
    TreeNode(...)]);  
    TreeNode (2, [TreeNode(...);  
    TreeNode(...)])]
```

10/18/12

33

## Infinite Recursive Values

```
# match lab_tree  
  with TreeNode (x, _) -> x;;  
- : int = 2
```

10/18/12

34

## Records

- Records serve the same programming purpose as tuples
- Provide better documentation, more readable code
- Allow components to be accessed by label instead of position
  - Labels (aka *field names*) must be unique)
  - Fields accessed by suffix dot notation

10/18/12

35

## Record Types

- Record types must be declared before they can be used in OCaml

```
# type person = {name : string; ss : (int * int  
  * int); age : int};;  
type person = { name : string; ss : int * int *  
  int; age : int; }
```

- person is the type being introduced
- name, ss and age are the labels, or fields

10/18/12

36

## Record Values

- Records built with labels; order does not matter

```
# let teacher = {name = "Elsa L. Gunter";  
age = 102; ss = (119,73,6244)};;  
val teacher : person =  
  {name = "Elsa L. Gunter"; ss = (119, 73,  
    6244); age = 102}
```

10/18/12

37

## Record Pattern Matching

```
# let {name = elsa; age = age; ss =  
  (_,_,s3)} = teacher;;  
val elsa : string = "Elsa L. Gunter"  
val age : int = 102  
val s3 : int = 6244
```

10/18/12

38

## Record Field Access

```
# let soc_sec = teacher.ss;;  
val soc_sec : int * int * int = (119,  
  73, 6244)
```

10/18/12

39

## Record Values

```
# let student = {ss=(325,40,1276);  
name="Joseph Martins"; age=22};;  
val student : person =  
  {name = "Joseph Martins"; ss = (325, 40,  
    1276); age = 22}  
# student = teacher;;  
- : bool = false
```

10/18/12

40

## New Records from Old

```
# let birthday person = {person with age =  
  person.age + 1};;  
val birthday : person -> person = <fun>  
# birthday teacher;;  
- : person = {name = "Elsa L. Gunter"; ss =  
  (119, 73, 6244); age = 103}
```

10/18/12

41

## New Records from Old

```
# let new_id name soc_sec person =  
  {person with name = name; ss = soc_sec};;  
val new_id : string -> int * int * int -> person  
  -> person = <fun>  
# new_id "Guieseppe Martin" (523,04,6712)  
  student;;  
- : person = {name = "Guieseppe Martin"; ss  
  = (523, 4, 6712); age = 22}
```

10/18/12

42